

NASA Contractor Report 172363

NASA-CR-172363
19840019610

COMPUTATION OF IMAGINARY-SIDE PRESSURE
DISTRIBUTIONS OVER THE FLEXIBLE WALLS
OF THE TEST SECTION INSERT FOR THE
0.3-M TRANSONIC CRYOGENIC TUNNEL

M. J. Goodyer

UNIVERSITY OF SOUTHAMPTON
Department of Aeronautics and Astronautics
Southampton, England
(Subcontractor)

KENTRON INTERNATIONAL, INC.
Kentron Technical Center
Hampton, Virginia 23666

Contract NAS1-16000
June 1984

LIBRARY COPY

JUN 20 1984

LANGLEY RESEARCH CENTER
LIBRARY, NASA
HAMPTON, VIRGINIA



National Aeronautics and
Space Administration

Langley Research Center
Hampton, Virginia 23665

NASA Technical Library



3 1176 01444 4476

CONTENTS

1. Introduction
 2. Program notes
 3. Listing
 4. Comparison with potential-flow test case
- Appendices

1. INTRODUCTION.

The object is to provide a method for determining the pressure distribution along the flexible walls of the 13 inch insert for the 0.3m Cryogenic Pressure Tunnel. The program IFLEX, the subject of this report, provides a method for general wall shapes. The program is based on incompressible potential flow theory, with the addition of an allowance for compressibility effects through the Prandtl-Glauert factor.

The program is quite compact, suitable for mini and micro computers, and is presented in BASIC language.

The essence of the method is to create mathematically an artificial flowfield which contains a streamline springing from infinity having the same shape as a wall. To one side of this streamline, corresponding to the imaginary-flow side of a flexible wall, there are no singularities, and the velocity distribution along the streamline is that which would exist in a semi-infinite inviscid flowfield over that side of the flexible wall. The input data is therefore the wall shape. The flowfield comprises a free stream influenced by a set of sources and sinks positioned regularly along a line parallel to the free stream. The source line is a tangent to the wall at its closest point to the axis of the test section. Often this tangent will in fact pass through the wall anchor point, jack zero, simply because the wall is everywhere else further away.

The input data comprises a set of wall deflections from "straight", measured at jacks. In the context of this work, "straight" must be interpreted as aerodynamically straight as defined in NASA CR-165936.

The strength of the source/sink set is adjusted until a streamline passes through the defined wall co-ordinates given

by the movement of jacks. The first source/sink is positioned $\frac{3}{4}$ " downstream of the anchor point (the upstream fixed end of the wall) the remainder being spaced at regular $1\frac{1}{2}$ " intervals downstream. The velocity/pressure coefficient is determined at "computing points" on the wall mid-way (measured streamwise) between sources, that is at regular $1\frac{1}{2}$ " intervals downstream from the anchor point.

The nature of the wall shape and jack spacing suggests the adoption of the following policy. Refer to test section drawing LD-525722 for relevant wall dimensions. There is a slope discontinuity built into the walls near to jack 19. The influence of the discontinuity on the static pressure at this jack station will cause the wall to be driven to an unrealistic position. The wall will be driven outwards in the subsonic flow normally expected in this region. It is suggested that in streamlining the walls (empty, or with a model present) the real-side pressure at jack 19 be ignored, and the jack be driven to position the wall at a point given by an extrapolation of the wall line through jacks 17 and 18. In this way the influence of the discontinuity on events at jack 18 will be minimised. Following this, the program to determine the pressures over the imaginary sides of the walls need only account for the movements of jacks 1 to 18 (jack 0 being the anchor point), with a small straight extrapolation towards the position of jack 19.

This arrangement results in the creation of many more sources and "computing points" than there are jacks, although many of the jacks do coincide with computing points particularly in the region of the model. Appendix 1 clarifies the positions of devices along a wall. The station measurements coincide with those on drawing LD-525722. The station of jack 0 is not definable exactly. It is taken as 0.5".

The two walls almost always have different shapes: they are therefore computed separately.

2. NOTES ON THE PROGRAM IFLEX.

The listing, given as Appendix 2, has many built-in comments aimed at clarifying the details of computation. These notes supplement that source of information.

The steps in the program comprise:

interpolation between jack displacements to give
wall displacements at each of 34 computing points
adjustment of source/sink strengths until a streamline
passes through the 34 points

computation of pressure coefficients at computing points
linear interpolation to give pressure coefficients at
jacks which lie between computing points.

Built into the program is a demonstration case: each wall is given a sinusoidal wave displaced upward. The wave is 20" long, $\frac{1}{2}$ " peak height. The imaginary flow above the top wall flows over what could be described as a $\frac{1}{2}$ " high "ridge" transverse to the free stream. As the wall is in some places not moved from its straight position, and is elsewhere raised above the straight, the sources lie along the "straight" position because this is also the tangent to the point (in this case in two areas) nearest to the tunnel axis.

The imaginary flow for the bottom wall traverses a "valley". The sources lie along a line displaced into the test section by the wave height, $\frac{1}{2}$ ". These geometries result in the appearance of a positive-negative-positive C_p sequence in the imaginary field outside of the top wall, with the strongest negative at the ridge crest. There is the opposite effect on the bottom wall. The output for this demonstration case is given in Appendix 3.

The program adopts linear extensions of the walls for $4\frac{1}{2}$ " beyond jack 18. The bend in the wall is 4.3" beyond jack 18.

4. A POTENTIAL-FLOW TEST CASE.

In order to check quantitatively on the output from the program, a potential-flow test case has been generated. The small program in Appendix 4 computes the shape and pressure

distribution along a streamline in the appropriate position for a wall. The streamline passes by a body created by a single source. A sketch is given on figure 1, which also has a plot of the shape of the streamline and its pressure distribution. The shape was used as input data to IFLEX, in the form of streamline deflections inline with each jack station. On the same plot is the exact pressure distribution compared with the predictions from IFLEX.

The two sets of pressure coefficient data are seen to be in close agreement. The maximum error in IFLEX is about ± 0.002 , likely to be compatible with the precision of the measurements of corresponding real-side pressures in the test section. One reason for IFLEX predicting a higher C_p near to the anchor point is the change in boundary slope which is assumed by IFLEX to occur near to the anchor point. No similar slope change occurs in the potential flow model.

The program is easily adapted in the manner indicated in its inbuilt comments to take as input data, if desired, the shapes of walls selected in wind tunnel tests. The computing time in BASIC on an HP 9845 is a total of 3 minutes for two walls including output as hard copy.

APPENDIX 1

IMAGINARY FLOWFIELD COMPUTATIONS FLEXIBLE WALL INSERT FOR 0.3m

POSITION TABLE

| Station inches | Jack# | Source# | Computing point |
|-------------------|-------|---------|--------------------|
| .50 | 0 | | 0 |
| 1.25 | | 0 | |
| 2.00 | | | 1 |
| 2.75 | | 1 | |
| 3.50 | | | 2 |
| 4.25 | | 2 | |
| 4.75 | 1 | | |
| 5.00 | | | 3 |
| 5.75 | | 3 | |
| 6.50 | | | 4 |
| 7.25 | | 4 | |
| 8.00 | | | 5 |
| 8.75 | | 5 | |
| 9.50 | | | 6 |
| 10.25 | | 6 | |
| 10.50 | 2 | | |
| 11.00 | | | 7 |
| 11.75 | | 7 | |
| 12.50 | | | 8 |
| 13.25 | | 8 | |
| 14.00 | | | 9 |
| 14.75 | | 9 | |
| 15.50 | 3 | | 10 |
| 16.25 | | 10 | |
| 17.00 | | | 11 |
| 17.75 | | 11 | |
| 18.50 | | | 12 |
| 19.25 | | 12 | |
| 19.50 | 4 | | |
| 20.00 | | | 13 |
| 20.75 | | 13 | |
| 21.50 | | | 14 |
| 22.25 | | 14 | |
| 22.50 | 5 | | |
| 23.00 | | | 15 |
| 23.75 | | 15 | |
| 24.50 | 6 | | 16 |
| 25.25 | | 16 | |
| 26.00 | 7 | | 17 |
| 26.75 | | 17 | |

| | | | |
|-------|----|----|----|
| 27.50 | 8 | | 18 |
| 28.25 | | 18 | |
| 29.00 | 9 | | 19 |
| 29.75 | | 19 | |
| 30.50 | 10 | | 20 |
| 31.25 | | 20 | |
| 32.00 | 11 | | 21 |
| 32.75 | | 21 | |
| 33.50 | 12 | | 22 |
| 34.25 | | 22 | |
| 35.00 | | | 23 |
| 35.50 | 13 | | |
| 35.75 | | 23 | |
| 36.50 | | | 24 |
| 37.25 | | 24 | |
| 37.50 | 14 | | |
| 38.00 | | | 25 |
| 38.75 | | 25 | |
| 39.50 | 15 | | 26 |
| 40.25 | | 26 | |
| 41.00 | | | 27 |
| 41.75 | | 27 | |
| 42.50 | 16 | | 28 |
| 43.25 | | 28 | |
| 44.00 | | | 29 |
| 44.75 | | 29 | |
| 45.50 | | | 30 |
| 46.25 | | 30 | |
| 46.50 | 17 | | |
| 47.00 | | | 31 |
| 47.75 | | 31 | |
| 48.50 | | | 32 |
| 49.25 | | 32 | |
| 50.00 | | | 33 |
| 50.75 | | 33 | |
| 51.50 | 18 | | 34 |

APPENDIX 2

```

10 ! RE-SAVED as "IFLEX"
20 PRINTER IS 0
30 OPTION BASE 0
40 PRINT
50 PRINT "          WALL IMAGINARY-SIDE PRESSURE COEFFICIENTS".
60 PRINT "          -----"
70 PRINT " FOR FLEXIBLE WALLED INSERT IN 0.3m CRYOGENIC WIND TUNNEL."
80 PRINT "          -----"
90 PRINT "          IFLEX"
100 PRINT "          August 6, 1982"
110 PRINTER IS 16
120 DIM Ty(18),By(18)      ! Jack displacements from aerodynamically straight,
130                        ! positive AWAY from test section centerline.
140 DIM Buf(18)           ! Buffer store for jack displacements.
150                        ! Dimensions are inches.
160 DIM Bufi(37)          ! Buffer store for interpolated wall displacements.
170 DIM A(37,37),S(36)    ! A for (source-wall angles)/PI, S for sources.
180 DIM Cp(36)            ! Pressure coefficients at computing points."
190 DIM X(18)             ! Streamwise stations of jacks.
200 DIM Di(18)            ! Used only in generating a set of prog. test data.
210 DATA 0.5,4.75,10.5,15.5,19.5,22.5,24.5,26,27.5,29,30.5,32,33.5,35.5,37.5
220 DATA 39.5,42.5,46.5,51.5      ! Jack stations.
230 Mo=0                    ! Reference Mach nr.
240 IF Mo<.9 THEN 280
250 INPUT "Mach may be too high. Input P[CONT] if to proceed:",P$
260 IF P$="P" THEN 280
270 GOTO 250
280 Beta=SQR(1-Mo*Mo)      ! Prandtl-Glauert factor.
290 FOR K=0 TO 18
300 READ X(K)
310 Di(K)=0
320 NEXT K
330 FOR J=4 TO 14 STEP 1 ! Jack nr.
340 Di(J)=.25+.25*SIN((X(J)-24.0)*PI/10) ! Sinusoid test case. Wave .5" high
350 NEXT J                ! and 20" wavelength centered on jack 9 at station 19".
360 FOR I=0 TO 18 STEP 1 ! Change of sign in lines 380 or 390 will exchange
370 ! ridge for valley.
380 Ty(I)=Di(I)          ! For top wall this wave is a ridge: negative Cp on crest.
390 By(I)=-Di(I)         ! " bottom " " " " " valley: pos. " in hollow
400 NEXT I                ! Test case formed.
410 ! The programme proper starts here.Ty() & By() must already be loaded.
420 Co=0                 ! Zero when computing top wall, & set to 1 for bottom wall.
430 GOTO 490
440 FOR Ii=0 TO 18
450 Buf(Ii)=By(Ii)       ! Load buffer store with bottom y-coordinates.
460 NEXT Ii
470 Yy=0
480 GOTO 520
490 FOR B=0 TO 18
500 Buf(B)=Ty(B)         ! " " " " " top "
510 NEXT B
520 X=.5                 ! X is to be raised from this value in 1.5" increments.
530 N1=0
540 Bufi(0)=0            ! Zero displacement at sta. 0.5 inch.
550 Y=0
560 N9=3                 ! First jack # which coincides with the chosen X-spacing.
570 Xc=X(N9)
580 FOR N=0 TO 15 STEP 1
590 X1=X(N)
600 X2=X(N+1)
610 X3=X(N+2)
620 X4=X(N+3)
630 Y1=Buf(N)
640 Y2=Buf(N+1)
650 Y3=Buf(N+2)
660 Y4=Buf(N+3)

```

```

670 GOSUB 2090
680 X=X+1.5 ! Next station at which interpolation to be made.
690 N1=N1+1 ! Counter for nr. of stations interpolated.
700 IF N<15 THEN 720 ! Go if not interpolating between jacks 16-18.
710 GOTO 730
720 IF X>X3 THEN 800 ! Most times interp. only in center patch.
730 Y=Y4+A2*(X-X4)^3+B2*(X-X4)^2+C2*(X-X4) ! Interpolated displacement.
740 IF X>51.5 THEN 1050 ! No interest beyond jack 18.
750 BuFi(N1)=Y ! Interpolated wall displacement.
760 IF Y<Yy THEN 900 ! Search for the minimum Y.
770 IF X=Xc THEN 830 ! Sta. X coincides with a jack.
780 IF X>Xc THEN 870 ! Bypassed a jack. Go to make Xc the next jack's sta.
790 GOTO 680
800 N1=N1-1
810 X=X-1.5
820 NEXT N
830 N9=N9+1
840 IF N9>18 THEN 920
850 Xc=X(N9)
860 GOTO 790
870 N9=N9+1 ! Jack counter.
880 Xc=X(N9)
890 GOTO 790
900 Yy=Y
910 GOTO 770
920 BuFi(35)=2*BuFi(34)-BuFi(33) ! One interval (1.5") extension in st. line.
930 BuFi(36)=2*BuFi(35)-BuFi(34) ! "
940 BuFi(37)=2*BuFi(36)-BuFi(35) ! "
950 FOR K1=0 TO 37
960 BuFi(K1)=BuFi(K1)-Yy ! Displacements are now all relative to a source
970 ! line passing through the point on the wall
980 ! closest to the centerline of the test section.
990 NEXT K1
1000 ! Start of aerodynamics.
1010 ! Linearized.
1020 FOR A1=0 TO 36
1030 S(A1)=BuFi(A1+1)-BuFi(A1) ! 34 approximate source half-strengths.
1031 S(A1)=S(A1)*(1+.75*BuFi(A1+1))
1040 NEXT A1
1050 PRINT
1060 IF Co=1 THEN 1090
1070 PRINT " Computing & correcting source/sink strengths for top wall:"
1080 GOTO 1110
1090 PRINTER IS 16
1100 PRINT " Computing source/sink strengths for bottom wall:"
1110 PRINT " - 2 iterations -"
1120 PRINT " Pass 1 (slow- 35 secs)"
1130 FOR P7=1 TO 2 ! Pass number, correcting sources.
1140 F6=-Yy
1150 FOR N=1 TO 37 STEP 1 ! Interval number.
1160 F2=BuFi(N) ! Free stream contribution to SF.
1170 FOR P=0 TO 36 ! Sources in turn.
1180 IF P=N-1 THEN 1250 ! Miss out the source just upstream.
1190 IF P7>1 THEN 1230
1200 T=BuFi(N)/(1.5*(N-P-.5))
1210 IF ABS(T)<.05 THEN 1430 ! Assume Tan(small angle) = angle,
1220 A(N,P)=ATN(T)/PI ! to speed up computation.
1230 F5=S(P)*A(N,P) ! SF contribution from this source.
1240 F2=F2+F5 ! Sum all source flow contributions to stream function SF.
1250 NEXT P
1260 IF P7>1 THEN 1300
1270 T=4*Bui(N)/3 ! Tangent of angle between point on wall & source just U/S
1280 IF ABS(T)<.05 THEN 1450
1290 A(N,N-1)=ATN(T)/PI
1300 S(N-1)=(F2-F6)/(1-A(N,N-1)) ! Adjust source to equalise SF's.
1310 F6=F6+S(N-1) ! SF for next N.
1320 NEXT N

```

```

1330 IF P7=5 THEN 1360
1340 PRINT "          Pass";P7+1;"    (11 secs)"
1350 NEXT P7
1360 GOTO 1470          ! Unless you would like the sources.
1370 PRINT
1380 PRINT "Station          Source strength          Source#"
1390 FOR Nh=0 TO 36
1400 PRINT USING 2060;Nh*1.5+1.25,2*S(Nh),Nh
1410 NEXT Nh
1420 GOTO 1470
1430 A(N,P)=T/PI
1440 GOTO 1230
1450 A(N,N-1)=T/PI
1460 GOTO 1300
1470 PRINT
1480 PRINT "          Have sources."
1490 PRINT "          Computing Cp's..."
1500 GOTO 1590          ! But there is the option of printing Cp at
1510                      ! 1.5" intervals by not going to 1590.
1520 PRINT
1530 IF Co=1 THEN 1560
1540 PRINT "          TOP WALL"
1550 GOTO 1570
1560 PRINT "          BOTTOM WALL"
1570 PRINT "Station","Cp"
1580 PRINT
1590 FOR Q=0 TO 34          ! Computing point. Station is .5+1.5*Q
1600 C7=0          ! See separate notes.
1610 C8=0
1620 FOR R=0 TO 36          ! Sources in turn.
1630 T1=(Q-R-.5)*1.5          ! Streamwise distance from source to point.
1640 T2=S(R)*T1
1650 T3=Bf1(Q)*Bf1(Q)+T1*T1
1660 T4=T2/T3
1670 C7=C7+T4
1680 T5=S(R)*Bf1(Q)
1690 T6=T5/T3
1700 C8=C8+T6
1710 NEXT R
1720 C7=C7/PI
1730 C8=C8/PI
1740 C9=(-2*C7-C7*C7-C8*C8)/Beta
1750 Cp(Q)=C9          ! Cp.
1760 GOTO 1780          ! But not if you want Cp at 1.5".
1770 PRINT USING 2070;Q*1.5+.5,C9
1780 NEXT Q
1790 PRINT
1800 PRINTER IS 0
1810 IF Co=1 THEN 1840
1820 PRINT "          Top wall"
1830 GOTO 1850
1840 PRINT "          Bottom wall"
1850 PRINT
1860 PRINT "          Jack          Cp"
1870 FOR E=1 TO 18          ! Cp at jacks by linear interpolation.
1880 X=X(E)
1890 FOR G=0 TO 34
1900 X1=.5+G*1.5
1910 IF X1>X THEN 1930
1920 NEXT G
1930 Cpa=Cp(G-1)
1940 Cpb=Cp(G)
1950 Cp=Cpa+(Cpb-Cpa)*(X-X1+1.5)/1.5
1960 PRINT USING 2080;E,Cp
1970 NEXT E
1980 PRINT

```

```

1990 PRINT
2000 IF Co=1 THEN 2030
2010 Co=1
2020 GOTO 440
2030 PRINT
2040 PRINT
2050 END
2060 IMAGE DD,DD,13X,D.DDDDD,13X,DD
2070 IMAGE DD,D,13X,D.DDDD
2080 IMAGE 10X, DD,6X,MDD.DDDD
2090 X5=X3-X4
2100 X6=X2-X4
2110 X7=X1-X4
2120 Y5=Y3-Y4
2130 Y6=Y2-Y4
2140 Y7=Y1-Y4
2150 B1=X5*X5-X5*X5*X5/X6
2160 B3=X7*X7-X7*X7*X7/X6
2170 C1=X5-X5*X5*X5/(X6*X6)
2180 C3=X7-X7*X7*X7/(X6*X6)
2190 Z1=Y5-Y6*(X5*X5*X5)/(X6*X6*X6)
2200 Z3=Y7-Y6*(X7*X7*X7)/(X6*X6*X6)
2210 C2=Z1*B3/B1-Z3
2220 C2=C2/(C1*B3/B1-C3)
2230 B2=(Z1-C2*C1)/B1
2240 A2=(Y5-B2*X5*X5-C2*X5)/(X5*X5*X5)
2250 RETURN

```

APPENDIX 3

WALL IMAGINARY-SIDE PRESSURE COEFFICIENTS

FOR FLEXIBLE WALLED INSERT IN 0.3m CRYOGENIC WIND TUNNEL.

IFLEX

August 6, 1982

Top wall

| Jack | Cp |
|------|--------|
| 1 | .0060 |
| 2 | .0108 |
| 3 | .0165 |
| 4 | .0747 |
| 5 | .0383 |
| 6 | -.0483 |
| 7 | -.1245 |
| 8 | -.1823 |
| 9 | -.2038 |
| 10 | -.1823 |
| 11 | -.1246 |
| 12 | -.0486 |
| 13 | .0395 |
| 14 | .0806 |
| 15 | .0555 |
| 16 | .0220 |
| 17 | .0126 |
| 18 | .0072 |

Bottom wall

| Jack | Cp |
|------|--------|
| 1 | -.0055 |
| 2 | -.0097 |
| 3 | -.0110 |
| 4 | -.0782 |
| 5 | -.0254 |
| 6 | .0656 |
| 7 | .1243 |
| 8 | .1605 |
| 9 | .1725 |
| 10 | .1605 |
| 11 | .1244 |
| 12 | .0659 |
| 13 | -.0261 |
| 14 | -.0841 |
| 15 | -.0543 |
| 16 | -.0188 |
| 17 | -.0114 |
| 18 | -.0064 |

APPENDIX 4

```

10  ! Potential flow streamline : source in uniform flow. Source strength
20  ! chosen such that body thickness is T" far downstream. The streamline
30  ! deflections which the programme gives apply to a line which passes
40  ! through a point 6.5" above the centerline (on which the source is
50  ! located, inline with jack 9) at station 0.5", that is at the anchor
60  ! point jack 0. Program POT. X-Y axis origin is at source.
70  DIM X(18)
80  INPUT "Thickness of body far downstream,inches",T
90  DATA 4.75,10.5,15.5,19.5,22.5,24.5,26,27.5,29,30.5,32,33.5,35.5,37.5
100 DATA 39.5,42.5,46.5,51.5
110 PRINT "          ";T;"INCH BODY."
120 PRINT "JACK      STATION      DEFLECTION      Cp"
130 FOR J=1 TO 18
140 READ X(J)
150 Y0=6.5
160 X=X(J)-29
170 F=6.5+ATN(6.5/-28.5)*T/(2*PI)
180 IF X=0 THEN 360
190 IF X<0 THEN 210
200 F=6.5+ATN(6.5/-28.5)*T/(2*PI)+T/2
210 FOR Y=Y0 TO 10 STEP .04
220 F1=Y+T*ATN(Y/X)/(PI*2)
230 IF F1>F THEN 250
240 NEXT Y
250 Y1=Y
260 FOR Y=Y1 TO 0 STEP -.001
270 F1=Y+T*ATN(Y/X)/(PI*2)
280 IF F1<F THEN 300
290 NEXT Y
300 Y1=Y
310 FOR Y=Y1 TO 10 STEP .0001
320 F1=Y+T*ATN(Y/X)/(PI*2)
330 IF F1>F THEN 370
340 NEXT Y
350 GOTO 370
360 Y=F+T/4      ! Y in closed form above origin.
370 R=X*X+Y*Y      ! R^2!
380 C=-(T*X)/(PI*R)-(T*X/(2*PI*R))^2-(T*Y/(2*PI*R))^2
390 PRINT USING 440;J,X+29,Y-Y0,C
400 NEXT J
410 PRINT
420 PRINT
430 END
440 IMAGE DD, 7X, DD.DD, 7X,DD.DDDD, 7X,DD.DDDDD

```

2 INCH BODY.

| JACK | STATION | DEFLECTION | Cp |
|------|---------|------------|---------|
| 1 | 4.75 | .0122 | .02433 |
| 2 | 10.50 | .0368 | .03033 |
| 3 | 15.50 | .0729 | .03767 |
| 4 | 19.50 | .1225 | .04434 |
| 5 | 22.50 | .1831 | .04644 |
| 6 | 24.50 | .2413 | .04207 |
| 7 | 26.00 | .2964 | .03277 |
| 8 | 27.50 | .3602 | .01731 |
| 9 | 29.00 | .4286 | -.00211 |
| 10 | 30.50 | .4959 | -.02063 |
| 11 | 32.00 | .5566 | -.03421 |
| 12 | 33.50 | .6083 | -.04191 |
| 13 | 35.50 | .6632 | -.04531 |
| 14 | 37.50 | .7049 | -.04440 |
| 15 | 39.50 | .7366 | -.04173 |
| 16 | 42.50 | .7714 | -.03698 |
| 17 | 46.50 | .8028 | -.03126 |
| 18 | 51.50 | .8284 | -.02576 |




Figure 1. Test case and comparison between exact and IFLEX values of pressure coefficient.

| | | | | | |
|---|--|-----------------------------|--|--|--|
| 1. Report No. NASA CR-172363 | | 2. Government Accession No. | | 3. Recipient's Catalog No. | |
| 4. Title and Subtitle Computation of Imaginary-Side Pressure Distributions Over the Flexible Walls of the Test Section Insert for the 0.3-m Transonic Cryogenic Tunnel | | | | 5. Report Date June 1984 | |
| | | | | 6. Performing Organization Code | |
| 7. Author(s) *M. J. Goodyer | | | | 8. Performing Organization Report No. | |
| | | | | 10. Work Unit No. | |
| 9. Performing Organization Name and Address Kentron International, Inc. Kentron Technical Center Hampton, VA 23666 | | | | 11. Contract or Grant No. NAS1-16000 - Task TZ-47 | |
| | | | | 13. Type of Report and Period Covered Contractor Report | |
| 12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, D.C. 20546 | | | | 14. Sponsoring Agency Code 505-31-53-10 | |
| | | | | | |
| 15. Supplementary Notes *University of Southampton Langley Technical Monitor: Charles L. Ladson Progress Report | | | | | |
| 16. Abstract Two dimensional airfoil testing in an adaptive wall test-section wind tunnel requires the computation of the imaginary flow fields extending outward from the top and bottom test section walls. A computer program was developed to compute the flow field which would be associated with an arbitrary test section wall shape. The program is based on incompressible flow theory with a Prandtl-Glauert compressibility correction. The program was validated by comparing the streamline and the pressure field generated by a source in uniform flow with the results from the computer program. A listing of the program, the validation test results, and a sample program are included. | | | | | |
| 17. Key Words (Suggested by Author(s)) 2-D Airfoils Adaptive Wall Wind Tunnels | | | 18. Distribution Statement Unclassified - Unlimited Star Category - 02 | | |
| 19. Security Classif. (of this report) Unclassified | 20. Security Classif. (of this page) Unclassified | 21. No. of Pages 15 | 22. Price A02 | | |



| LIBRARY MATERIAL SLIP | | |
|--|------|-----|
| DO NOT REMOVE SLIP FROM MATERIAL | | |
| <u>Delete your name from this slip when returning material to the library.</u> | | |
| NAME | DATE | MS |
| D. Chan | 7/04 | 267 |
| | | |
| | | |
| | | |
| | | |
| | | |
| | | |
| | | |
| | | |
| | | |
| | | |
| | | |